Лабораторна робота № 8 Задачі теорії масового обслуговування

Мета роботи: Оволодіти основними методами розв’язання та аналізу задач масового обслуговування для задач про багатоканальну систему з відмовами, задачу про багатоканальну систему з чергами та з обмеженою довжиною черги.

10.2 Завдання до лабораторної роботи №8

Визначити ймовірнісні характеристики зайнятості обслуговуючого персоналу СМО з відмовами за даними, що наведено в табл. 10.1.

Таблиця 10.1 – Індивідуальні варіанти до лабораторної роботи №8

№ варіанту	Щільність вхідного потоку заявок λ, клієнтів/хв.	Середній час обслуговування одного клієнта t_{cp}, хв.	Кількість обслуговуючого персоналу n, осіб
1	0,03	15	1,2,3
2	0,04	12	1,2,3
3	0,8	1	3,4,5
4	0,03	12	1,2,3
5	0,04	15	1,3,4
6	0,02	10	1,2,3
7	0,02	12	2,3,4
8	0,01	10	2,3,4
9	0,05	5	1,2,3
10	0,06	10	1,2,3
11	0,03	10	3,4,5
12	0,04	11	3,4,5
13	0,01	15	1,2,3
14	0,02	5	1,2,3
15	0,03	12	1,2,3
16	0,1	2	1,2,3
17	0,02	10	1,2,3
18	0,01	10	2,3,4
19	0,05	12	1,2,3
20	0,03	11	2,3,4
21	0,04	10	1,2,3
22	0,02	7	2,3,4
23	0,2	4	1,2,3
24	0,07	11	2,3,4
25	0,08	12	1,2,3
26	0,09	9	1,2,3
27	0,3	3	1,2,3
28	0,4	2	1,2,3
29	0,09	10	2,3,4
30	0,1	7	1,2,3

10.3 Методичні рекомендації до виконання лабораторної роботи №8

Приклад 10.1 Розглянемо задачу про *багатоканальн*у систему ***з відмовами***.

Фірма надає невідкладну допомогу з питань, пов'язаних з програмним забезпеченням комп'ютерної техніки. За рік у фірму звертається у середньому 2000 осіб за певним питанням. Фірма приймає клієнтів 6 год. на день протягом 255 днів на рік. Середній час обслуговування клієнта t_{cp}. Визначити ймовірність зайнятості обслуговуючого персоналу фірми, коли їх кількість $n = 1, 2, 3$.

Розв'язання. Обчислимо щільність вхідного потоку, тобто число клієнтів, що звертаються у фірму протягом хвилини:

$$\lambda = \frac{2000}{255 \cdot 6 \cdot 60} = 0{,}022 \text{ осіб/хв.}$$

Для подальших обчислень застосовуємо формули

$$\mu = \frac{1}{t_{cp}},\ \alpha = \frac{\lambda}{\mu},\ p_k = \frac{\frac{\alpha^k}{k!}}{\sum_{i=0}^{n} \frac{\alpha^i}{i!}}.$$

і проводимо розрахунки за допомогою таблиць Microsoft Excel (рис. 10.2 і 10.3).

	A	B	C	D	E	F
1		ДАНО				
2	λ	0,022				
3	tср	12				
4						
5		Розрахунок				
6	μ	=1/B3				
7	α	=B2/B6				
8						
9		i	0	1	2	3
10		α^i	=$B$7^C9	=$B$7^D9	=$B$7^E9	=$B$7^F9
11		i!	=FACT(C9)	=FACT(D9)	=FACT(E9)	=FACT(F9)
12		(α^i)/i!	=C10/C11	=D10/D11	=E10/E11	=F10/F11
13						
14	n	знаменник	p0	p1	p2	p3
15	1	=SUM(C12:D12)	=C$12/$B15	=D$12/$B15		
16	2	=SUM(C12:E12)	=C$12/$B16	=D$12/$B16	=E$12/$B16	
17	3	=SUM(C12:F12)	=C$12/$B17	=D$12/$B17	=E$12/$B17	=F$12/$B17

Рис. 10.2 – Програмування комірок на аркуші MS Excel задачі про багатоканальну систему з відмовами

	A	B	C	D	E	F
1		ДАНО				
2	λ	0,022				
3	tcp	12				
4						
5		Розрахунок				
6	μ	0,083333333				
7	α	0,264				
8						
9		i	0	1	2	3
10		α^i	1	0,264	0,069696	0,0184
11		i!	1	1	2	6
12		(α^i)/i!	1	0,264	0,034848	0,003067
13						
14	n	знаменник	p0	p1	p2	p3
15	1	1,264	0,791139	0,208861		
16	2	1,298848	0,769913	0,203257	0,02683	
17	3	1,301914624	0,7681	0,202778	0,026767	0,002355

Рис. 10.3 – Розрахунки для задачі про багатоканальну систему з відмовами

Як бачимо, збільшення числа працівників зменшує ймовірність відмови: для 1 працівника на фірмі ймовірність відмови дорівнює $p_1 = 0{,}20886$, для 2 працівників ймовірність відмови – $p_2 = 0{,}02683$, для 3 працівників – $p_3 = 0{,}00236$.

Для такого вхідного потоку навіть один працівник забезпечує високу ефективність роботи фірми.

Розглянемо, як змінюються ймовірнісні характеристики системи обслуговування при тій же продуктивності, але *при збільшенні вхідного потоку* в 10 разів, число звернень на рік тепер становить 20 000 осіб Тоді $\lambda = 0{,}22$ осіб/рік, а ймовірнісні характеристики обчислюємо за допомогою таблиць Microsoft Excel, змінивши в заготовленій раніше таблиці значення $\lambda = 0{,}022$ на $\lambda = 0{,}22$, отримаємо результат, продемонстрований на рис. 10.4.

	A	B	C	D	E	F
1		ДАНО				
2	λ	0,22				
3	tcp	12				
4						
5		Розрахунок				
6	μ	0,083333333				
7	α	2,64				
8						
9		i	0	1	2	3
10		α^i	1	2,64	6,9696	18,39974
11		i!	1	1	2	6
12		(α^i)/i!	1	2,64	3,4848	3,066624
13						
14	n	знаменник	p0	p1	p2	p3
15	1	3,64	0,274725	0,725275		
16	2	7,1248	0,140355	0,370537	0,489108	
17	3	10,191424	0,098122	0,259041	0,341935	0,300902

Рис. 10.4 – Розрахунки для задачі про багатоканальну систему з відмовами при збільшенні вхідного потоку в 10 разів

Отже,

для 1 працівника на фірмі ймовірність відмови дорівнює $p_1 = 0{,}725275$,

для 2 працівників ймовірність відмови – $p_2 = 0{,}489108$,

для 3 працівників – $p_3 = 0{,}300902$.

Зі збільшенням на порядок щільності вхідного потоку ймовірність відмови з ростом числа каналів не змінюється так різко, як у попередньому випадку.

Задамо число відвідувань у рік 200 000 осіб Тоді $\lambda = 2{,}2$ осіб/рік, а ймовірнісні характеристики обчислюємо допомогою таблиць Microsoft Excel, що наведена на рис. 10.5.

	A	B	C	D	E	F
1		ДАНО				
2	λ	2,2				
3	tcp	12				
4						
5		Розрахунок				
6	μ	0,083333333				
7	α	26,4				
8						
9		i	0	1	2	3
10		α^i	1	26,4	696,96	18399,74
11		i!	1	1	2	6
12		(α^i)/i!	1	26,4	348,48	3066,624
13						
14	n	знаменник	p0	p1	p2	p3
15	1	27,4	0,036496	0,963504		
16	2	375,88	0,00266	0,070235	0,927104	
17	3	3442,504	0,00029	0,007669	0,101229	0,890812

Рис. 10.5 – Розрахунки для задачі про багатоканальну систему з відмовами при збільшенні вхідного потоку в 100 разів

Для такого вхідного потоку і розглянутого числа каналів ймовірність відмови досить велика:

для 1 працівника на фірмі ймовірність відмови дорівнює $p_1 = 0{,}963504$,

для 2 працівників ймовірність відмови – $p_2 = 0{,}927104$,

для 3 працівників – $p_3 = 0{,}890812$.

Приклад 10.2 Розглянемо задачу про *багатоканальн*у систему ***з чергами*** за тих же умов, що й у попередньому прикладі.

Розв'язання. Для обчислень застосовуємо формули

$$p_k = \frac{\dfrac{\alpha^k}{k!}}{\sum_{i=0}^{n}\dfrac{\alpha^i}{i!} + \dfrac{\alpha^{n+1}}{n!\,(n-\alpha)}}, \quad 0 \le k \le n, m = \frac{\dfrac{n \cdot \alpha^{n+1}}{n!\,(n-\alpha)^2}}{\sum_{i=0}^{n}\dfrac{\alpha^i}{i!} + \dfrac{\alpha^{n+1}}{n!\,(n-\alpha)}},$$

$$p_ч = 1 - \sum_{k=1}^{n} p_k\,,\ p_h = \frac{\nu}{\lambda}\cdot m,\ \nu = \frac{1}{t_{cp}},$$

де $p_ч$ – ймовірність наявності черги.

Режим вважається установленим, якщо $\alpha < 1$. Для $N = 2000$ осіб/рік при будь-якому значенні n режим є установленим, а для $N = 20\ 000$ осіб/рік – тільки для $n \geq 3$. Відповідно до зазначених формул робимо розрахунки за допомогою таблиць Microsoft Excel $N = 2000$ осіб/рік (рис. 10.6 і 10.7).

	A	B	C	D	E	F	G
1		ДАНО					
2	λ	0,022					
3	tср	12					
4							
5		Розрахунок					
6	ν	=1/B3					
7	α	=B2/B6					
8							
9		i	0	1	2	3	
10		α^i	=$B$7^C9	=$B$7^D9	=$B$7^E9	=$B$7^F9	
11		i!	=FACT(C9)	=FACT(D9)	=FACT(E9)	=FACT(F9)	
12		(α^i)/i!	=C10/C11	=D10/D11	=E10/E11	=F10/F11	
13							
14	n	доданок	знаменник	p0	p1	p2	p3
15	1	=D12*$B$7/(A15-$B$7)	=SUM(C12:D12)+B15	=C$12/$C15	=D$12/$C15		
16	2	=E12*$B$7/(A16-$B$7)	=SUM(C12:E12)+B16	=C$12/$C16	=D$12/$C16	=E$12/$C16	
17	3	=F12*$B$7/(A17-$B$7)	=SUM(C12:F12)+B17	=C$12/$C17	=D$12/$C17	=E$12/$C17	=F$12/$C17
18							
19	n	чисельник	m	pч	ph		
20	1	=B15*A20/(A20-$B$7)	=B20/C15	=1-SUM(D15:E15)	=$B$6/$B$2*C20		
21	2	=B16*A21/(A21-$B$7)	=B21/C16	=1-SUM(D16:F16)	=$B$6/$B$2*C21		
22	3	=B17*A22/(A22-$B$7)	=B22/C17	=1-SUM(D17:G17)	=$B$6/$B$2*C22		

Рис. 10.6 – Програмування комірок на аркуші MS Excel задачі про багатоканальну систему з чергами

	A	B	C	D	E	F	G
1		ДАНО					
2	λ	0,022					
3	tср	12					
4							
5		Розрахунок					
6	ν	0,0833333					
7	α	0,264					
8							
9		i	0	1	2	3	
10		α^i	1	0,264	0,069696	0,0184	
11		i!	1	1	2	6	
12		(α^i)/i!	1	0,264	0,034848	0,003067	
13							
14	n	доданок	знаменни	p0	p1	p2	p3
15	1	0,0946957	1,358696	0,736	0,194304		
16	2	0,0052995	1,304147	0,766784	0,202431	0,026721	
17	3	0,0002959	1,302211	0,767925	0,202732	0,026761	0,002355
18							
19	n	чисельник	m	pч	ph		
20	1	0,1286626	0,094696	0,069696	0,358696		
21	2	0,0061054	0,004682	0,004064	0,017733		
22	3	0,0003245	0,000249	0,000227	0,000944		

Рис. 10.7 – Розрахунки для задачі про багатоканальну систему з чергами

Маємо: ймовірність наявності черги $p_ч$

при 1 працівникові на фірмі складатиме 0,0697,
при 2 працівниках – 0,00406,
при 3 – 0,00023,

ймовірність того, що заявка залишить систему не обслуговуваною p_h

при 1 працівникові на фірмі складатиме 0,3587,
при 2 працівниках – 0,01773,
при 3 – 0,00094,

середня довжина черги m

при 1 працівникові на фірмі складатиме 0,0947 осіб,
при 2 працівниках – 0,00468,
при 3 – 0,00025.

Аналогічно можна отримати, що для $N = 20\ 000$ осіб/рік при $n = 3$ (рис. 10.8).

	A	B	C	D	E	F	G
1		ДАНО					
2	λ	0,22					
3	tcp	12					
4							
5		Розрахунок					
6	v	0,0833333					
7	α	2,64					
8							
9		i	0	1	2	3	
10		α^i	1	2,64	6,9696	18,39974	
11		i!	1	1	2	6	
12		(α^i)/i!	1	2,64	3,4848	3,066624	
13							
14	n	доданок	знаменни	p0	p1	p2	p3
15	3	22,488576	32,68	0,0306	0,080783	0,106634	0,093838
16							
17	n	чисельник	m	pч	ph		
18	3	187,4048	5,734541	0,688145	2,172175		

Рис. 10.8 – Розрахунки для задачі про багатоканальну систему з чергами при $N=$ 20 000 осіб/рік

Маємо: ймовірність наявності черги $p_ч$=0,688; середня довжина черги m =5,735 осіб

Приклад 10.3 Розглянемо задачу СМО ***з обмеженою довжиною черги,*** що становить $j=6$ осіб, за тих же умов, що й у попередніх прикладах, у припущенні, що число відвідувань у рік 20 000 осіб

Розв'язання. Ймовірнісні характеристики обчислюємо за формулами

$$p_k=\frac{\dfrac{\alpha^k}{k!}}{\sum_{i=0}^{n}\dfrac{\alpha^i}{i!}+\dfrac{\alpha^n}{n!}\sum_{r=1}^{j}\left(\dfrac{\alpha}{n}\right)^r},\quad 0\le k\le n,\ p_{n+s}=\frac{\dfrac{\alpha^n}{n!}\left(\dfrac{\alpha}{n}\right)^s}{\sum_{i=0}^{n}\dfrac{\alpha^i}{i!}+\dfrac{\alpha^n}{n!}\sum_{r=1}^{j}\left(\dfrac{\alpha}{n}\right)^r},\quad 1\le s\le j.$$

за допомогою таблиць Microsoft Excel. Ті значення ймовірностей, що виділено на рис. 10.9 і 10.10 блакитним кольором, знайдено за другою з наведених формул.

	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1	ДАНО													
2	0,22													
3	12													
4														
5	Розрахунок													
6	=1/B3													
7	=B2/B6													
8														
9	i	0	1	2	3	4	5	6						
10	α^i	=$B$7^C9	=$B$7^D9	=$B$7^E9	=$B$7^F9									
11	i!	=FACT(C9)	=FACT(D9)	=FACT(E9)	=FACT(F9)									
12	(α^i)/i!	=C10/C11	=D10/D11	=E10/E11	=F10/F11									
13		(α/n)^i							сума 2					
14		=($B$7/$A14)^C$9	=($B$7/$A14)^D$9	=($B$7/$A14)^E$9	=($B$7/$A14)^F$9	=($B$7/$A14)^G$9	=($B$7/$A14)^H$9	=($B$7/$A14)^I$9	=SUM(D14:I14)					
15		=($B$7/$A15)^C$9	=($B$7/$A15)^D$9	=($B$7/$A15)^E$9	=($B$7/$A15)^F$9	=($B$7/$A15)^G$9	=($B$7/$A15)^H$9	=($B$7/$A15)^I$9	=SUM(D15:I15)					
16		=($B$7/$A16)^C$9	=($B$7/$A16)^D$9	=($B$7/$A16)^E$9	=($B$7/$A16)^F$9	=($B$7/$A16)^G$9	=($B$7/$A16)^H$9	=($B$7/$A16)^I$9	=SUM(D16:I16)					
17														
18	Сума 1	знаменник	p0	p1	p2	p3	p4	p5	p6	p7	p8	p9	q	Q
19	=SUM(C12:D12)	=B19+J14*D12	=C$12/$C19	=D$12/$C19	=$D$12*D$14/$C19	=$D$12*E$14/$C19	=$D$12*F$14/$C19	=$D$12*G$14/$C19	=$D$12*H$14/$C19	=$D$12*I$14/$C19			=1-K19	=N19*$B$2
20	=SUM(C12:E12)	=B20+J15*E12	=C$12/$C20	=D$12/$C20	=E$12/$C20	=$E$12*D$15/$C20	=$E$12*E$15/$C20	=$E$12*F$15/$C20	=$E$12*G$15/$C20	=$E$12*H$15/$C20	=$E$12*I$15/$C20		=1-L20	=N20*$B$2
21	=SUM(C12:F12)	=B21+J16*F12	=C$12/$C21	=D$12/$C21	=E$12/$C21	=F$12/$C21	=$F$12*D$16/$C21	=$F$12*E$16/$C21	=$F$12*F$16/$C21	=$F$12*G$16/$C21	=$F$12*H$16/$C21	=$F$12*I$16/$C21	=1-M21	=N21*$B$2

Рис. 10.9 – Програмування комірок на аркуші MS Excel для розрахунків задач СМО з обмеженою довжиною черги

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1		ДАНО													
2	λ	0,22													
3	tcp	12													
4															
5		Розрахунок													
6	v	0,083333													
7	α	2,64													
8															
9		i	0	1	2	3	4	5	6						
10		α^i	1	2,64	6,9696	18,39974									
11		i!	1	1	2	6									
12		(α^i)/i!	1	2,64	3,4848	3,066624									
13	n					(α/n)^i				сума 2					
14	1		1	2,64	6,9696	18,39974	48,57532	128,2389	338,5506	543,3741					
15	2		1	1,32	1,7424	2,299968	3,035958	4,007464	5,289853	17,69564					
16	3		1	0,88	0,7744	0,681472	0,599695	0,527732	0,464404	3,927703					
17															
18	n	Сума 1	знаменник	p0	p1	p2	p3	p4	p5	p6	p7	p8	p9	q	Q
19	1	3,64	1438,14763	0,000695	0,001836	0,004846	0,012794	0,033776	0,089169	0,235407	0,621476			0,378524	0,083275
20	2	7,1248	68,790576	0,014537	0,038377	0,050658	0,066869	0,088267	0,116512	0,153796	0,203011	0,267974		0,732026	0,161046
21	3	10,19142	22,2362134	0,044972	0,118725	0,156717	0,137911	0,121362	0,106798	0,093983	0,082705	0,07278	0,064047	0,935953	0,20591

Рис. 10.10 – Результати розрахунків задач СМО з обмеженою довжиною черги

Ймовірність того, що заявка залишить систему не обслуговуваною, дорівнює

при 1 працівникові на фірмі $p_h = p_{1+6} = p_7 = 0{,}62148$

при 2 працівниках $p_h = p_{2+6} = p_8 = 0{,}24797$

при 3 –$p_h = p_{3+6} = p_9 = 0{,}06405$.

Відносна пропускна спроможність в таблицях позначена $q = 1 - p_h$. Абсолютна пропускна спроможність $Q = \lambda q$, тобто число заявок, що обслуговуються за 1 хв. дорівнює

при 1 працівникові на фірмі $Q = 0{,}08328$осіб/хв.,

при 2 працівниках $Q = 0{,}16105$ осіб/хв.,

при 3 –$Q = 0{,}20591$ осіб/хв.

Питання для самоконтролю до теми 10 та лабораторної роботи №8

1. Що називають системою масового обслуговування (СМО)? Які основні елементи та ознаки СМО виділяють у теорії масового обслуговування?
2. Які типи систем масового обслуговування розрізняють залежно від конструкції обслуговуючого пристрою та дисципліни черги? Наведіть приклади одноканальних і багатоканальних СМО з відмовами та з чергами.
3. Які основні параметри використовуються для математичного опису СМО? Поясніть зміст інтенсивності вхідного потоку заявок λ, інтенсивності обслуговування μ та коефіцієнта завантаження α.
4. Які ймовірнісні характеристики СМО найчастіше визначають при аналізі системи? Що означають середня довжина черги, ймовірність відмови та пропускна спроможність системи?
5. За яких умов режим роботи СМО з чергою вважається установленим? Як змінюються ймовірність відмови та середня довжина черги зі зростанням інтенсивності вхідного потоку або кількості каналів обслуговування?
6. Поясніть принципи застосування імітаційного моделювання для проєктування систем масового обслуговування